brain boxes

IEEE 488 INTERFACE
C128 UPDATE



CBM 128/64 IEEE-488 INTERFACE UPDATE

VERSION 3.0 SUMMARY

Version 3.0 includes the following updates:-

1) Fix to allow VIZA CLASSIC and other VIZA programs to work in C128 mode.

2) Fix to small IEEE handshake bug, in listener to talker change over.

3) Change in initialisation routine to make the SRQ line default to an input on power up.

4) IEEE only i/o allowing IEEE only users to get a faster response on their devices, particularly disk drives. This is under user and program control.

5) Full set of IEEE Instrument control commands.
These commands include the IEEE standard commands like GTL, DCL etc
A full SERIAL POLL implemented.
A full PARALLEL POLL implemented.

These new commands make the C128 with BRAIN BOXES IEEE interface the ideal low cost laboratory instrument controller, so opening up a whole new field for sales.

A new edition of the IEEE manual will be available shortly incorporating the additional material in this book.

TABLE OF CONTENTS

CBM 128/64 IEEE-488 INTERFACE UPDATE

1) VIZA CLASSIC FIX.

Fix to allow VIZA CLASSIC and other VIZA programs to work in C128 mode. A bug that prevented these programs working was introduced accidentally in version 2.1. It has been detected and removed.

2) IEEE HANDSHAKE FIX.

Fix to IEEE handshake to assert ATN true (low) before returning NRFD and NDAC lines to a false (high) state when changing over from a listener to a talker. The IEEE-488 handshake is now perfectly correct! This bug did NOT effect Commodore peripheral equipment but only some laboratory IEEE instruments.

3) SRQ LINE ALWAYS INPUT CHANGE.

Slight change to initialisation routine to make the SRQ line default to an input on power up. This allows an easier SRQ test.
To test for an instrument asserting SRQ use the following BASIC program.

```
1 REM" SRQ PROGRAM
2 REM" IF A = 64 THEN SRQ HAS BEEN REQUESTED
3 REM
10 BANK(13) :REM " C128 ONLY!!!!
20 A=PEEK(57091) AND 64 :REM " PUT SRQ INTO A
30 B=PEEK(57090) :REM" RESET CB2 FOR NEXT SRQ
40 PRINT A
```

You can then go and do a full serial poll if you want using the following new routines.

4) FAST IEEE ENHANCEMENTS.

To allow IEEE users to get a faster response on their devices, particularly disk drives, it is now possible to prevent the slower serial bus routines from ever being called. The serial bus routines take much longer than the corresponding IEEE ones, the normal i/o has been done on a dual bus, handshaking both IEEE and serial bus together, for IEEE only users the serial bus timeout overhead can prove boring! So an IEEE only jump table has been added this allows you to send data to the IEEE 488 bus only.

You can make the C128 use these IEEE only calls and so perform IEEE only output in either of two ways.

a) FROM POWER UP
At power up time hold down the SHIFT key for C128 mode, hold down the SHIFT and CTRL keys for CP/M compatible mode.

b) UNDER PROGRAM CONTROL
Under program or user control by set calling the IEONLY routine.

```
POKE 5,0: BANK(13) :REM" SET UP THINGS
SYS 32897 :REM" SET IEEE ONLY
```

To resort to dual bus handshaking, which is the default method, allowing any mix of 1541, 1571 and IEEE devices, call DUAL.

```
POKE 5,0: BANK(13) :REM" SET UP THINGS
SYS 32900 :REM" SET IEEE ONLY

JUMP TABLE ENTRIES
32897 $8081 IEONLY Set IEEE only i/o.
32900 $8083 DUAL   Set DUAL serial/IEEE i/o.
```

IEEE ONLY JUMP TABLE

The existing Jump Table from $8000 to $807E, see manual page 17, has been extended to include the new IEEE only functions. The additions now include the normal i/o primitives (NSECND to NTKATN), indirect page three vector entry points and the full set of IEEE instrument control commands. Entry requirements and returns are as described in the Commodore Kernal specification, see C64 Programmers Reference Guide or C128 Programmers Reference Guide.

```
NEW JUMP TABLE ENTRIES
33066 $812A NSECND Send secondary addr to listener.
33069 $812D NTKSA  Send secondary addr to talker.
33072 $8130 NSETMO Set timeout byte.
33075 $8133 NACPTR Receive a byte from the bus.
33078 $8136 NCIOUT Send a byte to the bus.
33081 $8139 NUNTLK Untalk IEEE devices.
33084 $813C NUNLSN Unlisten IEEE devices.
33087 $813F NLISN  Command IEEE device to listen.
33090 $8142 NTALK  Command IEEE device to talk.
33093 $8145 NSCATN No sec. address to listener.
33096 $8148 NTKATN No sec. address to talker.
33099 $814B GRABIT MULTI USER ONLY!!!
33102 $814E DROPIT MULTI USER ONLY!!!
```

```
IEEE ONLY INDIRECT VECTORS FOR PAGE 3.
33105 $8151 KEY    IRQ vector.
33108 $8154 NTIMB  BREAK vector.
33111 $8157 XNMI   NMI vector.
33114 $815A NOPEN  OPEN vector.
33117 $815D NCLOSE CLOSE vector.
33120 $8160 NCHKIN CHECK IN vector.
33123 $8163 NCKOUT CHECK OUT vector.
33126 $8166 NCLRCH CLEAR CHANNEL vector.
33129 $8169 NBASIN INPUT vector.
33132 $816C NBSOUT OUTPUT vector.
33135 $816F INSTOP STOP KEY vector.
33138 $8172 NGETIN GET vector.
33141 $8175 NCLALL CLOSE ALL vector.
33144 $8178 EXMON  USER CMD vector.
33147 $817B NLOAD  LOAD vector.
33150 $817E NSAVE  SAVE vector.
```

To send one of these commands, UNLISTEN, from C128 BASIC 7 use the following code:-

```
5 POKE 5,0 :BANK(13)
20 SYS 33084 :REM" TELL IEEE BUS NOT TO LISTEN
```

5) IEEE INSTRUMENT CONTROLS.

INTRODUCTION
An IEEE Instrument control jump table has been added. This allows you to send addressed and universal commands to the IEEE 488 bus.
These commands include the IEEE standard commands like GTL, DCL etc
A full serial poll has been included as well as the stand alone calls SPE and SPD.
The PPC, PPD and the Parallel Poll routine are also implemented.

```
COMMAND JUMP TABLE
Addressed Commands
33153 $8181 Go To Local
33156 $8184 Selective Device Clear
33159 $8187 Parallel Poll Configure
33162 $818A Group Executive Trigger
33165 $818D Take Control

Universal Commands
33168 $8190 Local Lockout
33171 $8193 Device Clear
33174 $8196 Parallel Poll Unconfigure
33177 $8199 Serial Poll Enable
33180 $819C Serial Poll Disable
```

```
Poll Functions.
33183 $819F SERIAL POLL   A complete serial poll.
33186 $81A2 PARALLEL POLL A complete parallel poll.
33189 $81A5 Parallel Poll Enable.
33192 $81A8 Parallel Poll Disable.

User Command.
33195 $81AB Sends the byte in .A with ATN true.
```

PASSING PARAMETERS

Many instrument commands need parameters passing to them and several return data back to the user.
The format for passing data is:-

```
ON ENTRY
     .X register holds the device address
     .A register holds other information
     .Y register contents are discarded.
     The BASIC status byte ST is automatically zeroed on
     entry.

ON EXIT
     .X register, unchanged, holds the device address
     .A register, changed, holds any returned value
     .Y register contents are indeterminate.
     ST the BASIC status byte, holds information on the
     success of the current action. The meaning of these
     bits is as usual.
     bit7 set IEEE devices not present.
     bit6 set EOI on input. Last byte of data received.
     bit1 set Timeout on input. Device did not respond.
     bit0 set Timeout on output. Device did not respond.
```

From C128 BASIC 7 the values for the .A, .X and .Y registers can be passed with the SYS command.

SYS ADDRESS, .A, .X, .Y

or they can be POKE'd into reserved page zero memory locations.

```
     eg POKE 6,DATA sets the .A register
     eg POKE 7,DATA sets the .X register
     eg POKE 8,DATA sets the .Y register
```

The data that the IEEE routines return can be PEEKed from these same locations.

```
     eg PEEK(6) gives the .A register
     eg PEEK(7) gives the .X register
     eg PEEK(8) gives the .Y register
```

The status can be read by the BASIC command:- PRINT ST

From assembly language the ST byte is contained in the page zero memory location $90.

The BANK command is used in the C128 to select which bank of memory we want all PEEK, POKE and SYS to access. The IEEE rom is in the external cartridge low area, $8000- $C000. Since the IEEE rom uses the Kernal rom and page zero we need access to these memory locations too. This requires a value in the MMU register of $0A, a memory configuration equivalent to BANK(13).
In fact, we only need the BANK(13) statement ONCE ever in our program provided we do not issue any other BANK statements. i.e. If the first line of your program is BANK(13) you do not need to include any more BANK statements in your program.
In the following example BASIC 7 programs the bank routine is given each time for completeness only.

IEEE INSTRUMENT COMMANDS.

ADDRESSED COMMANDS.

Name	Byte		Hex	Decimal Address.
GTL	$01	Go To Local	8181	33153

PURPOSE. Returns responding devices to local control.
The GTL command is used to make a device respond to its own front panel controls. It is sometimes necessary for an operator to work from the front panel switches of the IEEE device for testing or trouble shooting purposes.
The device will return to remote (IEEE) control when it is addressed to listen again.

ENTRY REQUIREMENTS:
.X contains the device number.
EXIT RETURNS:
None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 X=6 :REM" TELL DEVICE 6 TO GO TO LOCAL
30 SYS 33153,0,X :REM" DUMMY PARAMETER IN THE .A REG
```

IEEE BUS ACTIVITY.

```
UNT
UNL
LAG
GTL
UNL (drop ATN)
```

Name	Byte		Hex	Decimal Address.
SDC	$04	Selective Device Clear	8184	33156

PURPOSE. Returns responding device to a pre-determined state.

The SDC command is used to make the particular addressed device reset itself. This allows you to put the particular device back to its initial power on state. Check your IEEE devices manual for the actual device dependent state.

All devices can be reset to their initialised state with the universal DCL command.

ENTRY REQUIREMENTS:

.X contains the device number.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 X=6 :REM" TELL DEVICE 6 TO CLEAR
30 SYS 33156,0,X :REM" DUMMY PARAMETER IN THE .A REG
```

IEEE BUS ACTIVITY.

```
UNT
UNL
LAG
SDC
UNL (drop ATN)
```

Name	Byte		Hex	Decimal Address.
GET	$08	Group Executive Trigger	818A	33162

PURPOSE. Initiates a pre-programmed action in the responding devices.

The GTL command is used to make the responding devices start doing a pre-programmed job. This may be outputing a voltage, taking a reading, starting an oscilloscope sweep etc. The actual job performed depends entirely on what your IEEE device is capable of and how you have pre-programmed it.

If X, the device address is greater than 30 (ie invalid IEEE address) then the GET statement is sent to all currently listening devices. So causing several devices to start their action together.

ENTRY REQUIREMENTS:

.X contains the device number or 31

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE. ONE DEVICE ONLY

```
10 BANK(13)
20 X=6 :REM" TELL DEVICE 6 TO TRIGGER
30 SYS 33162,0,X :REM" DUMMY PARAMETER IN THE .A REG
```

IEEE BUS ACTIVITY. ONE DEVICE ONLY

```
UNT
UNL
LAG
GET (drop ATN)
```

C128 BASIC 7 EXAMPLE. SEVERAL DEVICES

```
10 BANK(13)
11 POKE 186,4 : SYS 33087 :TELL DEVICE 4 TO LISTEN
12 POKE 186,5 : SYS 33087 :TELL DEVICE 5 TO LISTEN
13 POKE 186,6 : SYS 33087 :TELL DEVICE 6 TO LISTEN
........LISTEN AS MANY DEVICES AS NECESSARY
19 POKE 186,7 : SYS 33087 :TELL DEVICE 7 TO LISTEN
20 REM
30 X=31 :REM" INVALID DEVICE MEANS ALL LISTENERS G.E.T.
40 SYS 33162,0,X :REM" DUMMY PARAMETER IN THE .A REG..X
```

IEEE BUS ACTIVITY. SEVERAL DEVICES

```
LAG
LAG
LAG
...
LAG
GET (drop ATN)
```

Name	Byte		Hex	Decimal Address.
TCT	$09	Take Control	818D	33165

PURPOSE. The active controller transfers command to another device.

The TCT command is used to pass control of the IEEE bus to another controller. It is typically difficult to implement usefully since the C128 always expects to be the System Controller and the Active Controller. Any one needing more than the bare IEEE command please contact BRAIN BOXES on 051-220 2500.

ENTRY REQUIREMENTS:

.X contains the device number.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 X=6 :REM" TELL DEVICE 6 TO TAKE CONTROL
30 SYS 33165,0,X :REM" DUMMY PARAMETER IN THE .A REG
```

IEEE BUS ACTIVITY.

```
UNL
TAD
TCT (C128 drops ATN new controller asserts ATN)
```

UNIVERSAL COMMANDS.

Name	Byte		Hex	Decimal Address.
LLO	$11	Local Lockout	8190	33168

PURPOSE. Prevents devices acting under front panel or local control.

The LLO command is used to disable a device from responding to its own front panel controls. This is necessary when you want the instrument to be totally under the control of the IEEE controller (C128 computer). It prevents accidental pushing of the devices buttons interfering with its operation.

The device will return to local control when it receives the GTL command, or when the IEEE REN line goes false.

ENTRY REQUIREMENTS:

None.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 SYS 33168
```

IEEE BUS ACTIVITY.

LLO (drop ATN)

Name	Byte		Hex	Decimal Address.
DCL	$14	Device Clear	8193	33171

PURPOSE. Returns all responding device to a pre-determined state.

The DCL command is used to make all devices reset themselves. This allows you to put your devices back to their initial power on state. Not all IEEE devices can respond to this command however. Check your IEEE devices manual for the actual device dependent state.

Individual devices can be reset to their initialised state with the addressed SDC command.

ENTRY REQUIREMENTS:

None.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 SYS 33171
```

IEEE BUS ACTIVITY.

DCL (drop ATN)

SERIAL POLL COMMANDS.

Name	Byte		Hex	Decimal Address.
		Serial Poll Function	819F	33183
SPE	$18	Serial Poll Enable	8199	33177
SPD	$19	Serial Poll Disable	819C	33180

SERIAL POLL FUNCTION

PURPOSE. To determine which device has requested service from the controller and why.

The SERIAL POLL is usually called after a device performs a SERVICE REQUEST by setting SRQ low, see code on Page 3 to detect this. Serial poll is the only method that can correctly cause the device to return the SRQ false. In general, you will know which device requested the service and so will know which to poll. However, if it could be one of several devices, you should perform successive serial polls on the different devices until the status byte returned in .A indicates that you have found the guilty party.

NOTE: It is the device that causes the serial poll to be performed because it asserts SRQ when it needs servicing by the controller.

ON ENTRY .X holds the address we are polling to see if it requested service.

ON EXIT .X unchanged.
.A holds the device status byte. Bit6 set if it was the device requesting status, if so then the other bits will contain a code indicating the nature of the service requested.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 X=6 :REM" SEE IF DEVICE 6 REQUESTED SERVICE.
30 SYS 33183,0,X
40 AR=PEEK(6) :REM" GET THE SERVICE DATA INTO AR
50 AB=AR AND 64 :REM"IF AB=64 WE'VE GOT THE RIGHT DEVICE
```

IEEE BUS ACTIVITY.

```
UNT
UNL
SPE
TAD (drop ATN)
ACPTR (receive byte from device)
assert ATN
SPD
UNT (drop ATN)
```

Name	Byte		Hex	Decimal Address.
SPE	$18	Serial Poll Enable	8199	33177

PURPOSE. Initiates a serial poll.
The SPE command is used automatically in the SPOLL routine but it can be called separately to initiate your own serial poll routine. If you have, say, two devices that can request service, instead of serial polling them sequentially using SPOLL, you may wish to poll them together. You would then use SPE as part of your own routine.

Name	Byte		Hex	Decimal Address.
SPD	$19	Serial Poll Disable	819C	33180

PURPOSE. Disables serial poll mode.
The SPD command is used automatically in the SPOLL routine but it can be called separately to finish your own serial poll routine. It drops ATN on exit

ENTRY REQUIREMENTS:
None.
EXIT RETURNS:
None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
15 SYS 33081 :REM "UN TALK
20 SYS 33084 :REM "UN LISTEN
25 SYS 33177 :REM "SPE START SERIAL POLL
30 POKE 186,6 :SYS 33090 :REM "TELL DEVICE 6 TO TALK
35 SYS 33096 :REM "SEND NO SA TO TALKER drop ATN
40 SYS 33075 :REM "GET THE STATUS BYTE FROM DEVICE 6
45 A6=PEEK(6) :REM" GET THE SERVICE DATA INTO AR
50 B6=A6 AND 64 :REM"IF B6=64 WE'VE GOT THE RIGHT DEVICE
55 SYS 33081 :REM "UN TALK .
60 POKE 186,7 :SYS 33090 :REM "TELL DEVICE 7 TO TALK
65 SYS 33096 :REM "SEND NO SA TO TALKER drop ATN
70 SYS 33075 :REM "GET THE STATUS BYTE FROM DEVICE 7
75 A7=PEEK(6) :REM" GET THE SERVICE DATA INTO AR
80 B7=A7 AND 64 :REM"IF B7=64 WE'VE GOT THE RIGHT DEVICE
85 SYS 33081 :REM "UN TALK
90 SYS 33180 :REM "SPD FINISH SERIAL POLL MODE. drop ATN
```

IEEE BUS ACTIVITY.

```
UNT
UNL
SPE
TAD of first device (drop ATN)
ACPTR (receive byte from first device)
UNT
TAD of second device (drop ATN)
ACPTR (receive byte from second device)
UNT
SPD (drop ATN)
```

PARALLEL POLL COMMANDS.

Name	Byte		Hex	Decimal Address.
		Parallel Poll Function	81A2	33186
PPC	$05	Parallel Poll Configure	8187	33159
PPU	$15	Parallel Poll Unconfigure	8196	33174
PPE	$6X	Parallel Poll Enable	81A5	33189
PPD	$70	Parallel Poll Disable	81A8	33192

PARALLEL POLL FUNCTION

PURPOSE. Allows the controller to simultaneously determine the status of several devices.

Unlike the Serial Poll which is conducted after a device Requests Service, the Parallel Poll routine is initiated by the controller. It allows a whole group of devices to simultaneously return a status bit on the data lines. Which data line, and the sense of its response is determine by the controller when it send PPC, parallel poll configure. Devices power up in the PPU state and cannot respond to the Parallel Poll until after they have been configured.

The Parallel Poll makes good use of the controller's time since it can quickly check the state of 8 or more devices with one command.

The Commodore C128 knows which device is responsible for which data line because it has previously assigned those lines to each device using the PPC with PPE commands. Depending on the Parallel Poll Response the controller may take a variety of actions, chosen by the user's program.

ENTRY REQUIREMENTS:

None.

EXIT RETURNS:

.A contains the Parallel Poll response.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 SYS 33186
30 PR=PEEK(6) :REM" PUT THE RESPONSE BYTE INTO PR
```

IEEE BUS ACTIVITY.

ATN and EOI
at least 25 micro sec later read data lines in to .A
drop ATN and EOI

Name	Byte		Hex	Decimal Address.
PPC	$05	Parallel Poll Configure	8187	33159

PURPOSE. Allows the addressed listener to be assigned a data line for response to the parallel poll command.

The PPC command must be followed by a PPE command or a PPD command. The PPE command is used to tell the addressed listener which particular data line it is to use when responding to a parallel poll, the PPE command also tells the addressed device whether to assert the line when it requires service or to assert the line when it does not require service.

The PPD command is used to prevent the addressed device from responding to parallel polls. The previously assigned data lines is remembered.

The PPC command has been included for completeness only, it is advised that the PPE or PPD commands be used to configure a device for Parallel Poll since both these commands automatically send the PPC command.

ENTRY REQUIREMENTS:

None.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 SYS 33159
```

IEEE BUS ACTIVITY.

PPC keeps ATN true

Name	Byte		Hex	Decimal Address.
PPE	$6X	Parallel Poll Enable	81A5	33189

PURPOSE. Sends the complete PPC and PPE sequence to allow the addressed listener to be assigned a data line for response to the parallel poll command.

The PPC command is followed by a PPE command. The PPE command is used to tell the addressed listener which particular data line it is to use when responding to a parallel poll, the PPE command also tells the addressed device whether to assert the line when it requires service or to assert the line when it does not require service.

The PPE command has the bit value %0110sppp. Here s is the sense of response bit and ppp is the data line assigned to the device.

The sense information is in bit 3 (decimal 8).
If bit 3=0 then assert line when the device's individual status bit, IST, is zero.
If bit 3=1 then assert line when the device's IST is one.

The data line is assigned using bits 0-2.
If bits= 000, decimal 0, then data line 0
If bits= 001, decimal 1, then data line 1
If bits= 010, decimal 2, then data line 2
If bits= 011, decimal 3, then data line 3
If bits= 100, decimal 4, then data line 4
If bits= 101, decimal 5, then data line 5
If bits= 110, decimal 6, then data line 6
If bits= 111, decimal 7, then data line 7

The sense value and the data line value is logically OR'd together with the PPE command ($60) and sent to the bus after PPC with ATN true.

ENTRY REQUIREMENTS:
.X contains the device address.
.A contains the sense bit and data line assignments.

EXIT RETURNS:
None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 A= 8+2 :REM"ASSERT DATA LINE 2 TRUE FOR SERVICE
30 X=6 :REM" CONFIGURE DEVICE 6
40 SYS 33189,A,X
```

IEEE BUS ACTIVITY.

UNT
UNL
LAG
PPC
PPE includes line assignment and sense bit data
UNL (drop ATN)

Name	Byte		Hex	Decimal Address.
PPD	$70	Parallel Poll Disable	81A8	33192

PURPOSE. Sends the complete sequence to disable the addressed listener responding to the parallel poll command.

The PPC command is followed by a PPD command.

The PPD command is used to prevent the addressed device from responding to parallel polls. The previously assigned data lines is remembered.

ENTRY REQUIREMENTS:

.X contains the device address.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
30 X=6 :REM" STOP DEVICE 6 RESPONDING TO PARALLEL POLL
40 SYS 33192,0,X
```

IEEE BUS ACTIVITY.

```
UNT
UNL
LAG
PPC
PPD
UNL (drop ATN)
```

Name	Byte		Hex	Decimal Address.
PPU	$15	Parallel Poll Unconfigure	8196	33174

PURPOSE. Resets all devices with the Parallel Poll ability to the idle state.

The PPU command is used to make all devices that have the Parallel poll ability unable to respond to a parallel poll command.

For a device to respond to the Parallel Poll it must receive a PPC and PPE command. All devices power up in the PPU state. i.e they cannot respond to Parallel Poll.

ENTRY REQUIREMENTS:

None.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 SYS 33174
```

IEEE BUS ACTIVITY.

```
PPU (drop ATN)
```

Name	Byte		Hex	Decimal Address.
USER	$XX	User Command	81AB	33195

PURPOSE. To send a command byte to the IEEE bus.

The User Command is available for the user to send any bytes he wishes to the IEEE bus, with the ATN line asserted. This means that the byte is interpreted by the devices as a command rather than as a data byte. This command is the complement to NCIOUT, 33078, which simply sends a byte regardless of the state of the ATN line.

This routine can be used to send any of the standard IEEE commands by the user must ensure that he has got the sequence of commands and secondary commands correct.

Use the NSCATN routine, 33093, to drop ATN false again when you have finished sending your stream of data.

ENTRY REQUIREMENTS:

.A holds the byte the user wishes to send with ATN true.

EXIT RETURNS:

None.

C128 BASIC 7 EXAMPLE.

```
10 BANK(13)
20 SYS 33195,63 :REM " SEND UNLISTEN THIS WAY
```

IEEE BUS ACTIVITY.

assert ATN
USER COMMAND this case UNL (leaves ATN true)

COMMODORE 64 VERSION 3.0 UPDATE

1. The IEEE cartridge now works with the COMAL language cartridge.

   To boot COMAL:-
   a) Have the IEEE switched to C128 mode, not C64 mode.
   b) Insert the COMAL cartridge in the IEEE memory expansion slot.
   c) Switch the computer on.
   d) On power up the IEEE card will detect that another cartridge is present and then ask you the following question:
   PRESS S FOR SIMON BASIC
   PRESS C FOR COMAL

   Press C for COMAL.
   e) COMAL will now work with your IEEE devices and parallel printer.

2) The version 3.0 rom contains the same IEEE instrument commands as the C128 rom. The C64 calls start at $9F00 hex. Syntax and entry requirements are as for the C128 version. Except that the IEEE rom must be banked in to your system, do this as the very first command after you power on the C64, it will loose you 8k from BASIC's ram but 30K should be enough, ram at $C800 -CFFF is now free to use. Bank in the rom with:-

   SYS 51215 : POKE 56,128 : CLR

COMMAND JUMP TABLE

Addressed Commands

40704	$9F00	Go To Local
40707	$9F03	Selective Device Clear
40710	$9F06	Parallel Poll Configure
40713	$9F09	Group Executive Trigger
40716	$9F0C	Take Control

Universal Commands

40719	$9F0F	Local Lockout
40722	$9F12	Device Clear
40725	$9F15	Parallel Poll Unconfigure
40728	$9F18	Serial Poll Enable
40731	$9F1B	Serial Poll Disable

Poll Functions.

40734	$9F1E	SERIAL POLL A complete serial poll.
40737	$9F21	PARALLEL POLLA complete parallel poll.
40740	$9F24	Parallel Poll Enable.
40743	$9F27	Parallel Poll Disable.

User Command.

40746	$9F2A	Sends the byte in .A with ATN true.

PASSING PARAMETERS

Many instrument commands need parameters passing to them and several return data back to the user.
The format for passing data is:-

```
ON ENTRY
        .X register holds the device address
        .A register holds other information
        .Y register contents are discarded.
        The BASIC status byte ST is automatically zeroed
        on entry.

ON EXIT
        .X register, unchanged, holds the device address
        .A register, changed, holds any returned value
        .Y register contents are indeterminate.
        ST the BASIC status byte, holds information on the
        success of the current action. The meaning of
        these bits is as usual.
        bit7 set IEEE devices not present.
        bit6 set EOI on input. Last byte of data received.
        bit1 set Timeout on input. Device did not respond.
        bit0 set Timeout on output. Device did not
        respond.
```

From C64 BASIC 2 the values for the .A, .X and .Y registers can be POKE'd into reserved page three memory locations.

```
        eg POKE 780,DATA sets the .A register
        eg POKE 781,DATA sets the .X register
        eg POKE 782,DATA sets the .Y register
```

The data that the IEEE routines return can be PEEKed from these same locations.

```
        eg PEEK(780) gives the .A register
        eg PEEK(781) gives the .X register
        eg PEEK(782) gives the .Y register
```

The status can be read by the BASIC command:- **PRINT ST**